
Приложение 2

Программа моделирования работы стохастической системы

DECLARE FUNCTION lognormal! (mo!, sko!)

DECLARE FUNCTION binom! (n!, p!)

DECLARE FUNCTION puasson! (lam!)

DECLARE FUNCTION erlang! (k!, lam!)

DECLARE FUNCTION geometr! (p!)

DECLARE FUNCTION normal! (mo!, sko!)

DECLARE FUNCTION normal01! ()

DECLARE FUNCTION pokazat! (lam!)

DECLARE FUNCTION wejbull! (lam!, k!)

DECLARE FUNCTION ravnom! (a!, b!)

' Статистическое моделирование систем 

' Записывает в файл n значений случайной величины y=f(x,v) - функции

' от случайных величин x и v: y=v*(ln(x+1))^2

' x имеет распределение Эрланга 2-го порядка с параметром lam

' v имеет геометрическое распределение с параметром p

  p = .5

  lam = .1

  n = 100

  CLS

  OPEN "sta.txt" FOR OUTPUT AS #1

  'RANDOMIZE TIMER

  FOR i = 0 TO n - 1

      x = erlang(2, lam)

      v = geometr(p)

      y = v + LOG(x + 3) * LOG(x + 3)

      PRINT #1, USING "###.##  ###  ###.##"; x; v; y

  NEXT

  CLOSE #1

END

' Функция генерации случайных чисел, имеющих биномиальное распределение

FUNCTION binom (n, p)

     a = 1

     FOR m = 1 TO n

         a = a * (1 - p)

     NEXT

     s = 0

     FOR m = 0 TO n

         IF m <> 0 THEN a = a * (n - m + 1) * p / (m * (1 - p))

         s = s + a

         IF RND < s THEN GOTO 10

     NEXT

10 : binom = m

END FUNCTION

' Функция генерации случайных чисел, имеющих распределение Эрланга

FUNCTION erlang (k, lam)

      a = 0

      FOR i = 0 TO k - 1

          a = a + pokazat(lam)

      NEXT

      erlang = a

END FUNCTION

' Функция генерации случайных чисел, имеющих геометрическое распределение

FUNCTION geometr (p)

      geometr = INT(LOG(RND) / LOG(1 - p))

END FUNCTION

' Функция генерации логнормально распределенных случайных чисел

FUNCTION lognormal (mo, sko)

     lognormal = EXP(mo + sko * normal01)

END FUNCTION

' Функция генерации нормально распределенных случайных чисел

FUNCTION normal (mo, sko)

     normal = mo + sko * normal01

END FUNCTION

' Функция генерации нормально распределенных случайных чисел с парам.(0,1)

FUNCTION normal01

     ksi = SQR(-2 * LOG(RND))

     eta = 6.28318 * RND

     IF INT(2 * RND) = 0 THEN normal01 = ksi * COS(eta) ELSE normal01 = ksi * SIN(eta)

END FUNCTION

' Функция генерации случайных чисел распределенных по показательному закону

FUNCTION pokazat (lam)

    IF lam <> 0 THEN pokazat = -LOG(RND) / lam ELSE pokazat = 0

END FUNCTION

' Функция генерации случайных чисел, имеющих распределение Пуассона

FUNCTION puasson (lam)

     slu = RND

     m = 0: a = 1: s = 1

     IF slu <= s * EXP(-lam) THEN GOTO 20

     DO

        m = m + 1: a = a * lam / m: s = s + a

     LOOP WHILE slu > s * EXP(-lam)

20 : puasson = m

END FUNCTION

' Функция генерации равномерно распределенных случайных чисел на [a,b]

FUNCTION ravnom (a, b)

     ravnom = a + (b - a) * RND

END FUNCTION

' Функция генерации случайных чисел, распределенных по закону Вейбулла

FUNCTION wejbull (lam, k)

    IF lam <> 0 THEN wejbull = (-LOG(RND) / lam) ^ (1 / k) ELSE wejbull = 0

END FUNCTION

